
Cours VBA : les propriétés
Nous allons maintenant agir sur le contenu et l'apparence des cellules et des feuilles.
Commencez par ouvrir l'éditeur, ajoutez-y un module, copiez la macro ci-dessous et associez-la à un bouton formulaire (voir page "Sélections" en cas de besoin) :
Sub proprietes()
    'Macro incomplète
    Range ("A8")
End Sub
Nous voulons effectuer une action sur la cellule A8 avec ce début de macro.


Pour afficher la liste des possibilités que l'on peut associer à l'objet Range, ajoutez un point après Range ("A8") :
[image: liste - proprietes]
L'éditeur affiche alors les différentes possibilités ...
Pour ce premier exemple, cliquez sur "Value" puis sur la touche Tab pour valider ce choix.
Sub proprietes()
    'Macro incomplète
    Range("A8").Value
End Sub


La propriété Value est ici le contenu de la cellule.
Nous voulons maintenant donner la valeur 48 à A8 :
Sub proprietes()
    'A8 = 48
    Range("A8").Value = 48
    'Traduction :
    'La valeur de la cellule A8 est égale à 48
End Sub
Puis, la valeur Exemple de texte à A8 (important : le texte doit être mis entre " ") :
Sub proprietes()
    'A8 = Exemple de texte
    Range("A8").Value = "Exemple de texte"
End Sub
Dans ce cas, c'est bien la cellule A8 de la feuille où est lancée la procédure (ici, grâce au bouton formulaire) qui sera modifiée. Si vous créez un second bouton sur la feuille 2, ce sera alors la cellule A8 de la feuille 2 qui sera modifiée.
Pour modifier la cellule A8 de la feuille 2 en cliquant sur le bouton de la feuille 1, il faut ajouter avant Range : Sheets("Nom_de_la_feuille") ou Sheets(Numéro_de_la_feuille).
Sub proprietes()
    'A8 de la feuille 2 = Exemple de texte
    Sheets("Feuil2").Range("A8").Value = "Exemple de texte"
    'Ou :
    'Sheets(2).Range("A8").Value = "Exemple de texte"
End Sub
De même, si l'on souhaite modifier la cellule A8 de la feuille 2 d'un autre classeur ouvert, il faut ajouter avant Sheets et Range : Workbooks("Nom_du_fichier").
Sub proprietes()
    'A8 de la feuille 2 du classeur 2 = Exemple de texte
    Workbooks("Classeur2.xlsx").Sheets("Feuil2").Range("A8").Value = "Exemple de texte"
End Sub


Bien que Value ait été utilisé pour illustrer ces différents exemples, il n'est pas nécessaire de l'indiquer, car c'est automatiquement la valeur de la cellule qui est modifiée si rien n'est précisé.
Ces 2 lignes offrent un résultat identique :
Range("A8").Value = 48
Range("A8") = 48
Effacer le contenu de cellules
[image: liste2 - proprietes]
Sub proprietes()
    'Effacer le contenu de la colonne A
    Range("A:A").ClearContents
End Sub
Mise en forme du texte
Après avoir ajouté Font., la liste des propriétés que l'on peut attribuer à la mise en forme du texte apparaît :
[image: liste3 - proprietes]
La modification des couleurs sera détaillée à la page suivante ...
Mise en forme : taille du texte
Sub proprietes()
    'Modifier la taille du texte des cellules A1 à A8
    Range("A1:A8").Font.Size = 18
End Sub
Mise en forme : texte en gras
Sub proprietes()
    'Mettre en gras les cellules A1 à A8
    Range("A1:A8").Font.Bold = True
End Sub
Bold = True signifie Caractères en gras = Oui.
Pour supprimer la mise en forme "caractères gras" à un texte, il faut donc remplacer "Oui" par "Non", autrement dit, "True" par "False" :
Sub proprietes()
    'Enlever la mise en forme "gras" des cellules A1 à A8
    Range("A1:A8").Font.Bold = False
End Sub
Mise en forme : texte en italique
Sub proprietes()
    'Mettre en italique les cellules A1 à A8
    Range("A1:A8").Font.Italic = True
End Sub
Mise en forme : texte souligné
Sub proprietes()
    'Souligner les cellules A1 à A8
    Range("A1:A8").Font.Underline = True
End Sub

Mise en forme : police0
Sub proprietes()
    'Modifier la police de caractères des cellules A1 à A8
    Range("A1:A8").Font.Name = "Arial"
End Sub
Ajouter des bordures
[image: liste4 - proprietes]
Sub proprietes()
     'Ajouter une bordure aux cellules A1 à A8
     Range("A1:A8").Borders.Value = 1
     'Value = 0    => pas de bordure
End Sub
Modifier la mise en forme de la sélection actuelle
Sub proprietes()
     'Ajouter une bordure aux cellules sélectionnées
     Selection.Borders.Value = 1
End Sub
Modifier les propriétés d'une feuille
Sub proprietes()
      'Masquer une feuille
     Sheets("Feuil3").Visible = 0
      'Visible = -1     => annule l'effet
End Sub
N'oubliez pas que seule une toute petite minorité des possibilités de personnalisation sont indiquées ici.
Si la propriété dont vous avez besoin n'est pas détaillée ici, n'ayez pas peur de la rechercher grâce à la liste de choix et l'aide Excel.
L'enregistreur de macro peut également vous éviter de longues recherches. En enregistrant la manipulation dont vous avez besoin, vous pourrez retrouver plus facilement la propriété recherchée pour ensuite l'utiliser dans votre macro.
Modifier la valeur d'une cellule en fonction d'une autre
Nous voulons ici que A7 prenne la valeur de A1 :
[image: valeur - proprietes]
Nous allons donc demander que A7 prenne la valeur de A1, ce qui nous donne :
Sub proprietes()
     'A7 = A1
     Range("A7") = Range("A1")
     'Ou :
     'Range("A7").Value = Range("A1").Value
End Sub
Si nous voulons copier la taille du texte uniquement, le code serait :
Sub proprietes()
     Range("A7").Font.Size = Range("A1").Font.Size
End Sub
Ce qui est à gauche du signe = prend la valeur de ce qui est à droite du signe =.
Modifier la valeur d'une cellule en fonction de sa propre valeur
Nous allons créer ici un compteur de clics.
A chaque clic, la valeur de A1 sera augmentée de 1 :
Sub proprietes()
     'Compteur de clics en A1
     Range("A1") = Range("A1") + 1
End Sub
Excel exécute le code ligne par ligne, ces commentaires devraient vous aider à mieux comprendre ce même code :
'Pour exemple : avant l'exécution du code, A1 vaut 0

Sub proprietes()

    'Un clic a été fait sur le bouton, nous entrons dans la procédure
    'Pour le moment A1 vaut encore 0
    
    'PENDANT l'exécution de la ligne ci-dessous A1 vaut toujours 0
    Range("A1") = Range("A1") + 1 'Le calcul est alors : Nouvelle_valeur_de_A1 = 0 + 1
    'A1 vaut alors 1 seulement APRES l'exécution de la ligne de code
    
End Sub
With





Ce code permet de définir différentes propriétés à la cellule active :
Sub proprietes()
    ActiveCell.Borders.Weight = 3
    ActiveCell.Font.Bold = True
    ActiveCell.Font.Size = 18
    ActiveCell.Font.Italic = True
    ActiveCell.Font.Name = "Arial"
End Sub
Nous pouvons utiliser With pour éviter les répétitions d'ActiveCell dans le cas présent.
Voici comment With fonctionne :
Sub proprietes()
    'Début de l'instruction avec : WITH
    With ActiveCell
        .Borders.Weight = 3
        .Font.Bold = True
        .Font.Size = 18
        .Font.Italic = True
        .Font.Name = "Arial"
    'Fin de l'instruction avec : END WITH
    End With
End Sub
ActiveCell n'est donc plus répété.
Bien que ce ne soit pas indispensable dans ce cas, il est également possible d'éviter les répétitions de .Font, ce qui nous donnerait :
Sub proprietes()
    With ActiveCell
        .Borders.Weight = 3
        With .Font
            .Bold = True
            .Size = 18
            .Italic = True
            .Name = "Arial"
        End With
    End With
End Sub

Nous allons commencer par attribuer une couleur au texte en A1.
Après avoir ajouté Font., nous obtenons :
[image: color - couleurs]

Nous avons 2 possibilités pour définir la couleur : ColorIndex et ses 56 couleurs ou Color qui nous permettra d'utiliser n'importe quelle couleur.
ColorIndex
Voici les 56 couleurs disponibles avec ColorIndex :
[image: colorindex - couleurs]
Pour donner à notre texte l'une des 56 couleurs, nous écrirons :
Sub couleurs()
    'Couleur du texte en A1 : vert (Couleur n°10)
    Range("A1").Font.ColorIndex = 10
End Sub
Ce qui nous donne :
[image: test2 - couleurs]
Pour les versions d'Excel inférieures à 2007 : l'utilisation de ColorIndex est préférable à Color.
Color
Voici un exemple similaire avec Color :
Sub couleurs()
    'Couleur du texte en A1 : RGB(50, 200, 100)
    Range("A1").Font.Color = RGB(50, 200, 100)
End Sub
La couleur ici est : RGB(50, 200, 100)
RGB en français signifie Rouge-Vert-Bleu (RVB), les valeurs vont de 0 à 255 pour chaque couleur.
Quelques exemples de couleurs pour mieux comprendre :
· RGB(0, 0, 0) : noir
· RGB(255, 255, 255) : blanc
· RGB(255, 0, 0) : rouge
· RGB(0, 255, 0) : vert
· RGB(0, 0, 255) : bleu
Heureusement pour nous, il existe différentes solutions qui nous permettent de trouver facilement les valeurs RGB de la couleur qui nous intéresse.
Vous trouverez par exemple une liste de valeurs RGB sur la page suivante : liste de valeurs RGB


Pour donner une couleur violette à notre texte, nous pouvons donc rechercher les valeurs RGB de cette couleur sur la liste et écrire :
Sub couleurs()
    'Couleur du texte en A1 : RGB(192, 32, 255)
    Range("A1").Font.Color = RGB(192, 32, 255)
End Sub
Ce qui nous donne :
[image: test - couleurs]
Pour les versions d'Excel inférieures à 2007 : le nombre de couleurs est limité (la couleur disponible la plus proche de la valeur RGB sera utilisée).
Créer une bordure colorée
Nous allons créer une macro qui va ajouter une bordure à la cellule active avec ActiveCell.
La bordure sera rouge et épaisse :
Sub couleurs()
    'Epaisseur de la bordure
    ActiveCell.Borders.Weight = 4
    'Couleur de la bordure : rouge
    ActiveCell.Borders.Color = RGB(255, 0, 0)
End Sub
Aperçu :
[image: bordure - couleurs]
Pour appliquer cet effet à plusieurs cellules à la fois, nous pouvons utiliser Selection :
Sub couleurs()
    'Epaisseur de la bordure
    Selection.Borders.Weight = 4
    'Couleur de la bordure : rouge
    Selection.Borders.Color = RGB(255, 0, 0)
End Sub
Colorer le fond des cellules sélectionnées
Sub couleurs()
    'Colorer le fond des cellules sélectionnées
    Selection.Interior.Color = RGB(174, 240, 194)
End Sub
Aperçu :
[image: fond - couleurs]
Colorer l'onglet d'une feuille
Sub couleurs()
    'Colorer l'onglet de la feuille "Feuil1"
    Sheets("Feuil1").Tab.Color = RGB(255, 0, 0)
End Sub
Aperçu :
[image: onglet - couleurs]

Les variables permettent de stocker toutes sortes de données.
Voici un premier exemple :
'Affichage de la valeur de la variable dans une boîte de dialogue
Sub variables()
    'Déclaration de la variable
    Dim ma_variable As Integer
    'Attribution d'une valeur à la variable
    ma_variable = 12
    'Affichage de la valeur de ma_variable dans une MsgBox
    MsgBox ma_variable
End Sub


Cette première ligne de code est la déclaration de la variable (généralement placée en début de procédure).
Dim ma_variable As Integer
· Dim : déclaration de la variable
· ma_variable : nom choisi pour cette variable (sans espaces)
· As : déclaration du type de la variable
· Integer : type de la variable
Déclarer ses variables n'est pas obligatoire mais recommandé. Cela permet de s'y retrouver plus facilement, peut aider à résoudre plus facilement les problèmes, etc. Bref, mieux vaut prendre l'habitude de déclarer ses variables correctement.
Le type de la variable indique la nature de son contenu (texte, nombres, date, etc.).
Une valeur est ensuite donnée à cette variable :
ma_variable = 12
Et enfin, la valeur de la variable est affichée dans une boîte de dialogue :
MsgBox ma_variable
MsgBox "valeur" affiche une valeur dans une boîte de dialogue de la façon la plus simple.
Les boîtes de dialogue seront détaillées dans quelques leçons ...
Le résultat de ce code :
[image: msgbox - variables]
Si pour le moment vous ne comprenez pas bien l'intérêt d'utiliser des variables, soyez rassuré, les exemples abordés au cours des prochaines leçons vous en démontreront l'utilité ...
Les types de variables
	Nom
	Type
	Détails
	Symbole

	Byte
	Numérique
	Nombre entier de 0 à 255.
	

	Integer
	Numérique
	Nombre entier de -32'768 à 32'767.
	%

	Long
	Numérique
	Nombre entier de - 2'147'483'648 à 2'147'483'647.
	&

	Currency
	Numérique
	Nombre à décimale fixe de -922'337'203'685'477.5808 à 922'337'203'685'477.5807.
	@

	Single
	Numérique
	Nombre à virgule flottante de -3.402823E38 à 3.402823E38.
	!

	Double
	Numérique
	Nombre à virgule flottante de -1.79769313486232D308 à 1.79769313486232D308.
	#

	String
	Texte
	Texte.
	$

	Date
	Date
	Date et heure.
	

	Boolean
	Boolean
	True (vrai) ou False (faux).
	

	Object
	Objet
	Objet Microsoft.
	

	Variant
	Tous
	Tout type de données (type par défaut si la variable n'est pas déclarée).
	


Quelques exemples avec différents types :
'Exemple : nombre entier
Dim nbEntier As Integer
nbEntier = 12345
    
'Exemple : nombre à virgule
Dim nbVirgule As Single
nbVirgule = 123.45

'Exemple : texte
Dim varTexte As String
varTexte = "Excel-Pratique.com"

'Exemple : date
Dim varDate As Date
varDate = "06.02.2011"

'Exemple : vrai/faux
Dim varBoolean As Boolean
varBoolean = True
    
'Exemple : objet (objet Worksheet pour cet exemple)
Dim varFeuille As Worksheet
Set varFeuille = Sheets("Feuil2") 'Set => attribution d'une valeur à une variable objet
    
'Exemple d'utilisation de la variable objet : activation de la feuille
varFeuille.Activate
Les symboles indiqués dans le tableau ci-dessus permettent de raccourcir les déclarations de variables.
Par soucis de lisibilité, ils ne seront pas utilisés dans les leçons mais voici tout de même un exemple :
Dim exemple As Integer
Dim exemple%
Ces deux lignes sont identiques.
Il est possible de forcer les déclarations de variables en plaçant Option Explicit tout au début du module (une erreur sera ainsi générée en cas d'oubli de déclaration).
Exemple pratique
Nous allons maintenant créer par étapes une macro qui va récupérer le nom dans la cellule A2, le prénom dans la cellule B2, l'âge dans la cellule C2 et qui va les afficher dans une boîte de dialogue.
Fichier source : exercice_variables.xls
Nous commençons par déclarer les variables (sur la même ligne, séparées par des virgules) :
Sub variables()
    'Déclaration des variables
    Dim nom As String, prenom As String, age As Integer
    
End Sub
Nous attribuons ensuite les valeurs aux variables avec Cells :
Sub variables()
    'Déclaration des variables
    Dim nom As String, prenom As String, age As Integer
    
    'Valeurs des variables
    nom = Cells(2, 1)
    prenom = Cells(2, 2)
    age = Cells(2, 3)
    
End Sub








Et enfin, nous affichons le résultat dans la boîte de dialogue en concaténant les valeurs avec & (comme sous Excel).
Sub variables()
    'Déclaration des variables
    Dim nom As String, prenom As String, age As Integer
    
    'Valeurs des variables
    nom = Cells(2, 1)
    prenom = Cells(2, 2)
    age = Cells(2, 3)
    
    'Boîte de dialogue
    MsgBox nom & " " & prenom & ", " & age & " ans"
End Sub
Ce qui nous donne :
[image: resultat - variables]
Nous allons maintenant rechercher à afficher dans la boite de dialogue la ligne du tableau correspondant au numéro indiqué dans la cellule F5.
Voici l'objectif :
[image: objectif - variables]
La solution :
Sub variables()
    'Déclaration des variables
    Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
        
    'Valeurs des variables
    numero_ligne = Range("F5") + 1
    nom = Cells(numero_ligne, 1)
    prenom = Cells(numero_ligne, 2)
    age = Cells(numero_ligne, 3)
    
    'Boîte de dialogue
    MsgBox nom & " " & prenom & ", " & age & " ans"
End Sub
Ajout d'une variable :
'Déclaration des variables
Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
La variable numero_ligne prend ensuite la valeur de la cellule F5 à laquelle nous ajoutons 1 (pour ne pas tenir compte de la première ligne qui contient les titres du tableau) ainsi, numero_ligne aura pour valeur le n° de ligne des cellules qui nous intéressent :
numero_ligne = Range("F5") + 1
Il ne reste plus qu'à remplacer les n° de ligne des Cells par notre variable :
nom = Cells(numero_ligne, 1)
prenom = Cells(numero_ligne, 2)
age = Cells(numero_ligne, 3)
Notre macro affiche maintenant la ligne du tableau qui nous intéresse.
[image: objectif2 - variables]
Notez au passage que nous pouvons réduire cette procédure entière sur une ligne :
Sub variables()
    MsgBox Cells(Range("F5")+1,1) & " " & Cells(Range("F5")+1,2) & ", " & Cells(Range("F5")+1,3) & " ans"
End Sub
Le code fonctionne correctement, il est néanmoins beaucoup moins lisible que le précédent et plus difficile à retravailler (les codes ne seront donc pas réduits dans les leçons afin d'en faciliter la compréhension).

Les tableaux
Les variables permettent de stocker une valeur par variable, les tableaux permettent de stocker une multitude de valeurs (leur utilisation est proche de celle des variables).
Voici quelques exemples de déclarations :
'Exemple de déclaration de variable
Dim var1 As String
    
'Exemple de déclaration de tableau à 1 dimension
Dim tab1(4) As String
    
'Exemple de déclaration de tableau à 2 dimensions
Dim tab2(4, 3) As String
    
'Exemple de déclaration de tableau à 3 dimensions
Dim tab3(4, 3, 2) As String
Le tableau à 1 dimension :
'Exemple de déclaration de tableau à 1 dimension
Dim tab1(4) As String
[image: 1dim - variables suite]Dans cette déclaration, il n'y a qu'un chiffre entre parenthèses, il s'agit donc d'un tableau à une dimension. Ce chiffre indique la dernière case du tableau. tab1(4) est un tableau dont les cases vont de 0 à 4, il s'agit donc d'un tableau de 5 cases :
'Attribution de valeurs aux 5 cases
tab1(0) = "Valeur de la case 0"
tab1(1) = "Valeur de la case 1"
tab1(2) = "Valeur de la case 2"
tab1(3) = "Valeur de la case 3"
tab1(4) = "Valeur de la case 4"

La première case d'un tableau est le 0.
Autre exemple, le tableau à 2 dimensions :
'Exemple de déclaration de tableau à 2 dimensions
Dim tab2(4, 3) As String

[image: 2dim - variables suite]'Attribution de valeurs aux 3 cases colorées
tab2(0, 0) = "Valeur de la case rouge"
tab2(4, 1) = "Valeur de la case verte"
tab2(2, 3) = "Valeur de la case bleue"
Les constantes
Les constantes permettent de stocker des valeurs comme les variables, à la différence qu'on ne peut pas les modifier (d'où leur nom).
Par exemple, nous pouvons ajouter une constante pour éviter les répétitions de 6.87236476641 :
Sub exemple_const()
    Cells(1, 1) = Cells(1, 2) * 6.87236476641
    Cells(2, 1) = Cells(2, 2) * 6.87236476641
    Cells(3, 1) = Cells(3, 2) * 6.87236476641
    Cells(4, 1) = Cells(4, 2) * 6.87236476641
    Cells(5, 1) = Cells(5, 2) * 6.87236476641
End Sub
Cela facilite la lecture du code (en particulier pour des codes importants) et facilite le changement (manuel) de la valeur de la constante en cas de besoin :
Sub exemple_const()
    'Déclaration de la constante + attribution de sa valeur
    Const TAUX_ANNUEL As Double = 6.87236476641
    
    Cells(1, 1) = Cells(1, 2) * TAUX_ANNUEL
    Cells(2, 1) = Cells(2, 2) * TAUX_ANNUEL
    Cells(3, 1) = Cells(3, 2) * TAUX_ANNUEL
    Cells(4, 1) = Cells(4, 2) * TAUX_ANNUEL
    Cells(5, 1) = Cells(5, 2) * TAUX_ANNUEL
End Sub
La portée des variables
Si la variable est déclarée au début d'une procédure (Sub), elle ne peut être utilisée que dans cette même procédure. La valeur de la variable n'est pas conservée après l'exécution de la procédure.
Sub procedure1()
   Dim var1 As Integer
   ' => Utilisation de la variable dans la procédure uniquement
End Sub

Sub procedure2()
   ' => Impossible d'utiliser var1 ici
End Sub
Pour pouvoir utiliser une variable dans toutes les procédures d'un module, il suffit de la déclarer en début de module. De plus, cela permet de conserver la valeur de la variable jusqu'à la fermeture du classeur.
Dim var1 As Integer

Sub procedure1()
   ' => Utilisation de var1 possible
End Sub


Sub procedure2()
   ' => Utilisation de var1 possible
End Sub
Même principe pour utiliser une variable dans tous les modules, à la différence près que Dim est remplacé par Global :
Global var1 As Integer
Pour conserver la valeur d'une variable à la fin d'une procédure, remplacez Dim par Static :
Sub procedure1()
    Static var1 As Integer
End Sub
Pour conserver les valeurs de toutes les variables d'une procédure, ajoutez Static devant Sub :
Static Sub procedure1()
    Dim var1 As Integer
End Sub
Créer son propre type de variable
Nous n'allons pas nous attarder sur ce point, voici juste un exemple :
'Création d'un type de variable
Type invites
    nom As String
    prenom As String
End Type
    
Sub variables()
    'Déclaration
    Dim p1 As invites
    
    'Attributions des valeurs à p1
    p1.nom = "Smith"
    p1.prenom = "John"
    
    'Exemple d'utilisation
    MsgBox p1.nom & " " & p1.prenom
End Sub

les conditions
Les conditions sont très utiles en programmation, elles nous serviront à effectuer des actions en fonction de critères précis (même principe que la fonction SI).
La principale fonction est If, voici comment elle fonctionne :
If [CONDITION ICI] Then ' => SI condition validée ALORS
    'Instructions si vrai
Else ' => SINON
    'Instructions si faux
End If


Passons directement à la pratique et reprenons l'exemple développé à la leçon sur les variables. Il avait pour but d'afficher dans une boîte de dialogue la ligne du tableau correspondant au numéro indiqué dans la cellule F5. :
Fichier source : conditions.xls
[image: conditions - conditions]
Si vous entrez une lettre en F5, cela génère un bug. Nous voulons éviter cela.
Sub variables()
    'Déclaration des variables
    Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
        
    'Valeurs des variables
    numero_ligne = Range("F5") + 1
    nom = Cells(numero_ligne, 1)
    prenom = Cells(numero_ligne, 2)
    age = Cells(numero_ligne, 3)
    
    'Boîte de dialogue
    MsgBox nom & " " & prenom & ", " & age & " ans"
End Sub





Nous allons commencer par ajouter une condition pour vérifier que la valeur de la cellule F5 est numérique avant d'exécuter le code.
La fonction IsNumeric sera utilisée dans cette condition :
Sub variables()

    'SI la valeur entre parenthèses (cellule F5) est numérique (DONC SI CONDITION VRAIE) alors on exécute les instructions placées après THEN
    If IsNumeric(Range("F5")) Then
    
        'Déclaration des variables
        Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
        'Valeurs des variables
        numero_ligne = Range("F5") + 1
        nom = Cells(numero_ligne, 1)
        prenom = Cells(numero_ligne, 2)
        age = Cells(numero_ligne, 3)
        'Boîte de dialogue
        MsgBox nom & " " & prenom & ", " & age & " ans"
        
    End If
    
End Sub
Ajoutons également des instructions pour le cas où la condition n'est pas remplie :
Sub variables()

    If IsNumeric(Range("F5")) Then 'SI CONDITION VRAIE
    
        'Déclaration des variables
        Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
        'Valeurs des variables
        numero_ligne = Range("F5") + 1
        nom = Cells(numero_ligne, 1)
        prenom = Cells(numero_ligne, 2)
        age = Cells(numero_ligne, 3)
        'Boîte de dialogue
        MsgBox nom & " " & prenom & ", " & age & " ans"
   
     
    Else 'SI CONDITION FAUSSE
    
        'Boîte de dialogue : avertissement
        MsgBox "L'entrée " & Range("F5") & " n'est pas valide !"
        'Suppression du contenu de la cellule F5
        Range("F5").ClearContents
    
    End If
    
End Sub
Les valeurs non numériques ne sont désormais plus un problème.
Notre tableau contient 16 lignes de données, nous allons donc vérifier que la variable numero_ligne soit : "plus grande ou égale à 2" et "plus petite ou égale à 17".
Mais avant, voici les opérateurs de comparaison :
	=
	est égal à

	<>
	est différent de

	<
	est plus petit que

	<=
	est plus petit ou égal à

	>
	est plus grand que

	>=
	est plus grand ou égal à


Ainsi que d'autres opérateurs utiles :
	AND
	et
	[condition1] AND [condition2]
Les 2 conditions doivent être vraies

	OR
	ou
	[condition1] OR [condition2]
Au moins 1 des 2 conditions doit être vraie

	NOT
	faux
	NOT [condition1]
La condition doit être fausse


Ajoutons maintenant les conditions indiquées un peu plus haut en utilisant AND ainsi que les opérateurs de comparaison détaillés ci-dessus :
Sub variables()
    If IsNumeric(Range("F5")) Then 'SI NUMERIQUE
        Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
        numero_ligne = Range("F5") + 1

        If numero_ligne >= 2 And numero_ligne <= 17 Then 'SI N° CORRECT
            nom = Cells(numero_ligne, 1)
            prenom = Cells(numero_ligne, 2)
            age = Cells(numero_ligne, 3)
            MsgBox nom & " " & prenom & ", " & age & " ans"
        Else 'SI N° INCORRECT
            MsgBox "L'entrée " & Range("F5") & " n'est pas un numéro valide !"
            Range("F5").ClearContents
        End If
        
    Else 'SI NON NUMERIQUE
        MsgBox "L'entrée " & Range("F5") & " n'est pas valide !"
        Range("F5").ClearContents
    End If
End Sub
Pour rendre notre macro plus pratique, nous pouvons encore remplacer 17 par une variable contenant le nombre de lignes. Cela permettra d'ajouter/retirer des lignes à notre tableau sans avoir à modifier à chaque fois cette limite.
Pour cela, créons une variable nb_lignes et ajoutons cette fonction:
[image: nb lignes - conditions]








WorksheetFunction.CountA ne vous dit probablement rien mais il s'agit en fait de la fonction NBVAL que vous connaissez probablement déjà (sinon, cliquez ici).
Nous demandons à cette fonction de comptabiliser le nombre de cellules non vides de la première colonne et nous remplaçons ensuite 17 par nb_lignes :
Sub variables()
    If IsNumeric(Range("F5")) Then 'SI NUMERIQUE
        Dim nom As String, prenom As String, age As Integer, numero_ligne As Integer
        Dim nb_lignes As Integer
        
        numero_ligne = Range("F5") + 1
        nb_lignes = WorksheetFunction.CountA(Range("A:A")) 'Fonction NBVAL
        
        If numero_ligne >= 2 And numero_ligne <= nb_lignes Then 'SI N° CORRECT
            nom = Cells(numero_ligne, 1)
            prenom = Cells(numero_ligne, 2)
            age = Cells(numero_ligne, 3)
            MsgBox nom & " " & prenom & ", " & age & " ans"
        Else 'SI N° INCORRECT
            MsgBox "L'entrée " & Range("F5") & " n'est pas un numéro valide !"
            Range("F5").ClearContents
        End If

    Else 'SI NON NUMERIQUE
        MsgBox "L'entrée " & Range("F5") & " n'est pas valide !"
        Range("F5").ClearContents
    End If
End Sub
ElseIf
ElseIf permet d'ajouter plusieurs conditions à la suite :
If [CONDITION 1] Then ' => SI condition 1 validée ALORS
    'Instructions 1
ElseIf [CONDITION 2] Then ' => SINON, SI condition 2 validée ALORS
    'Instructions 2
Else ' => SINON
    'Instructions 3
End If
Si la condition 1 est vraie, les instructions 1 sont exécutées puis nous sortons de l'instruction If (qui débute avec If et finit à End If). Si la condition 1 est fausse, nous passons à la condition 2. Si celle-ci est vraie les instructions 2 sont exécutées si ce n'est pas le cas les instructions 3 seront alors exécutées.
Voici un exemple, avec en A1 une note de 1 à 6 (sans virgules pour cet exemple) et en B1 un commentaire en fonction de la note :
Sub commentaires_notes()
    'Variables
    Dim note As Integer, commentaire As String
    note = Range("A1")
    
    'Commentaire en fonction de la note
    If note = 6 Then
        commentaire = "Excellent résultat !"
    ElseIf note = 5 Then
        commentaire = "Bon résultat"
    ElseIf note = 4 Then
        commentaire = "Résultat satisfaisant"
    ElseIf note = 3 Then
        commentaire = "Résultat insatisfaisant"
    ElseIf note = 2 Then
        commentaire = "Mauvais résultat"
    ElseIf note = 1 Then
        commentaire = "Résultat exécrable"
    Else
        commentaire = "Aucun résultat"
    End If
    
    'Commentaire en B1
    Range("B1") = commentaire
End Sub
[image: notes - conditions]

Select
Une alternative aux instructions If contenant beaucoup ElseIf existe : Select, cette instruction étant plus adaptée dans ce genre de situations.
Voici la même macro avec Select :
Sub commentaires_notes()
    'Variables
    Dim note As Integer, commentaire As String
    note = Range("A1")
    
    'Commentaire en fonction de la note
    Select Case note    ' <= la valeur à tester (ici, la note)
    Case Is = 6         ' <= si la valeur = 6
        commentaire = "Excellent résultat !"
    Case Is = 5         ' <= si la valeur = 5
        commentaire = "Bon résultat"
    Case Is = 4         ' <= si la valeur = 4
        commentaire = "Résultat satisfaisant"
    Case Is = 3         ' <= si la valeur = 3
        commentaire = "Résultat insatisfaisant"
    Case Is = 2         ' <= si la valeur = 2
        commentaire = "Mauvais résultat"
    Case Is = 1         ' <= si la valeur = 1
        commentaire = "Résultat exécrable"
    Case Else           ' <= si la valeur n'est égale à aucune des valeurs ci-dessus
        commentaire = "Aucun résultat"
    End Select
    
    'Commentaire en B1
    Range("B1") = commentaire
End Sub
Notez que nous pouvons également utiliser les autres opérateurs de comparaison, par exemple :
Case Is >= 6         'si la valeur >= 6


Exemples avec plusieurs valeurs :
Case Is = 6, 7       'si la valeur = 6 ou 7
Case Is <> 6, 7      'si la valeur est différente de 6 ou 7
Case 6 To 10         'si la valeur = de 6 à 10
Condition en fonction d'un type
IsNumeric (fonction vue à la page précédente) renvoie TRUE (vrai) si la valeur est numérique et FALSE (faux) si ce n'est pas le cas :
If IsNumeric(Range("A1")) = True Then 'SI LA VALEUR EST NUMERIQUE ...
Le code suivant est identique au premier (il n'est pas nécessaire d'indiquer = True puisque que l'on cherche automatiquement à savoir si la condition est vraie) :
If IsNumeric(Range("A1")) Then 'SI LA VALEUR EST NUMERIQUE ...
Dans le cas où nous voulons vérifier si la valeur n'est pas numérique, nous avons également deux possibilités :
If IsNumeric(Range("A1")) = False Then 'SI LA VALEUR N'EST PAS NUMERIQUE ...
If Not IsNumeric(Range("A1")) Then 'SI LA VALEUR N'EST PAS NUMERIQUE ...
D'autres fonctions proches de IsNumeric :
If IsDate(Range("A1")) Then 'SI LA VALEUR EST UNE DATE ...
If IsEmpty(Range("A1")) Then 'SI VIDE ...
If var_objet Is Nothing Then 'SI OBJET NON INITIALISE ...
Condition en fonction du type d'une variable
Pour effectuer des actions en fonction du type d'une variable (Variant), nous aurons besoin de la fonction VarType.
Après avoir ajouté le signe =, la liste des types apparaît :
[image: vartype - conditions suite]
If VarType(ma_variable) = vbInteger Then 'SI ma_variable est de type Integer ...
La valeur des constantes :
	Constante
	Valeur

	vbEmpty
	0

	vbNull
	1

	vbInteger
	2

	vbLong
	3

	vbSingle
	4

	vbDouble
	5

	vbCurrency
	6

	vbDate
	7

	vbString
	8

	vbObject
	9

	vbError
	10


If VarType(ma_variable) = vbInteger Then 'SI ma_variable est de type Integer ...
'Identique à :
If VarType(ma_variable) = 2 Then 'SI ma_variable est de type Integer ...
Condition en fonction de la comparaison de 2 chaînes de caractères
Jusque-là nous n'avons vu que cela :
ma_variable = "Exemple 12345"

If ma_variable = "Exemple 12345" Then ' => VRAI
Les 2 chaînes de caractères sont identiques ici, rien d'extraordinaire ...
Maintenant, si nous voulons vérifier que la variable contienne bien la valeur "12345" sans tenir compte des autres caractères, nous utiliserons l'opérateur Like ainsi que * devant et derrière la valeur à rechercher.
Le caractère * peut remplacer : aucun, un ou plusieurs caractères :
ma_variable = "Exemple 12345"

If ma_variable Like "*12345*" Then ' => VRAI
Le caractère # peut remplacer un caractère numérique de 0 à 9 :
ma_variable = "Exemple 12345"

If ma_variable Like "Exemple 12###" Then ' => VRAI
Le caractère ? peut remplacer un caractère quelconque :
ma_variable = "Exemple 12345"

If ma_variable Like "?xemple?1234?" Then ' => VRAI
Nous pouvons également remplacer un caractère en fonction d'une plage de caractères ou de caractères précis :
· [abc] remplace un des caractères suivants : a b c
· [a-g] remplace un des caractères suivants : a b c d e f g
· [369] remplace un des caractères suivants : 3 6 9
· [2-5] remplace un des caractères suivants : 2 3 4 5
· [?*#] remplace un des caractères suivants : ? * #
ma_variable = "Exemple 12345"

If ma_variable Like "[BIEN]xemple 1234[4-7]" Then ' => VRAI
Pour remplacer un caractère non compris dans les valeurs entre crochets, un ! doit être ajouté après [ :
ma_variable = "Exemple 12345"

If ma_variable Like "[!FAUX]xemple 1234[!6-9]" Then ' => VRAI
Un caractère en majuscule n'est pas égal à ce même caractère en minuscule. Pour ne pas faire de distinctions entre majuscules-minuscules, placez Option Compare Text en début de module.

les boucles
While
Les boucles permettent de répéter des instructions un certain nombre de fois, ce qui peut être un gain de temps considérable.
Le code suivant numérote les cellules de la colonne A (de la ligne 1 à 12) :


Sub boucle_while()

    Cells(1, 1) = 1
    Cells(2, 1) = 2
    Cells(3, 1) = 3
    Cells(4, 1) = 4
    Cells(5, 1) = 5
    Cells(6, 1) = 6
    Cells(7, 1) = 7
    Cells(8, 1) = 8
    Cells(9, 1) = 9
    Cells(10, 1) = 10
    Cells(11, 1) = 11
    Cells(12, 1) = 12

End Sub
Ce code est très répétitif ...
Maintenant, imaginez qu'il faille numéroter plusieurs centaines de lignes ... Vous comprenez donc l'intérêt de créer des boucles.
Voici la boucle While :
Sub boucle_while()

    While [condition]
        'Instructions
    Wend

End Sub
Tant que la condition est vraie, les instructions sont exécutées en boucle (attention à ne pas créer une boucle infinie).
Voici la macro répétitive ci-dessus avec la boucle While :
Sub boucle_while()

    Dim numero As Integer
    numero = 1 'Numéro de départ (correspond ici au n° de ligne et au n° de numérotation)

    While numero <= 12 'TANT QUE la variable numero est <= 12, la boucle est répétée
        Cells(numero, 1) = numero 'Numérotation
        numero = numero + 1 'Le numéro est augmenté de 1 à chaque boucle
    Wend
        
End Sub
Avec cette boucle, si nous voulons numéroter 500 lignes, il suffit alors de remplacer 12 par 500 ...
Do Loop
Cette boucle fonctionne de la même manière que While Wend (tant que la condition est vraie, la boucle est exécutée) :
Sub boucle_do_while()

    Do While [condition]
        'Instructions
    Loop

End Sub
La condition peut également être placée en fin de boucle Do Loop, ce qui implique que les instructions sont exécutées au moins une fois :
Sub boucle_do_while()

    Do
        'Instructions
    Loop While [condition]

End Sub
Plutôt que de répéter la boucle tant que la condition est vraie, il est possible de quitter la boucle lorsque la condition est vraie en remplaçant While par Until :
Sub boucle_do_while()

    Do Until [condition]
        'Instructions
    Loop

End Sub
For
Sub boucle_for()

    For i = 1 To 5
        'Instructions
    Next

End Sub
La boucle For est répétée ici 5 fois.
A chaque répétition de la boucle, la variable i est automatiquement augmentée de 1 :
Sub boucle_for()

    For i = 1 To 5
        MsgBox i
    Next

End Sub
Quitter une boucle prématurément
Il est possible de quitter une boucle For prématurément grâce à l'instruction suivante :
Exit For 'Quitter une boucle For
En voici un exemple :
Sub boucle_for()

    Dim max_boucles As Integer
    max_boucles = Range("A1") 'En A1 : une limite de répétitions de la boucle est définie

    For i = 1 To 7 'Boucles prévues : 7
        If i > max_boucles Then 'Si A1 est vide ou contient un nombre < 7, diminution du nb de boucles
            Exit For 'Si condition vraie, on quitte la boucle For
        End If
    
       MsgBox i
    Next

End Sub
Les autres instructions Exit :
Exit Do 'Quitter une boucle Do Loop
Exit Sub 'Quitter une procédure
Exit Function 'Quitter une fonction
Télécharger ce cours au format PDF

les boucles (exercice)
Pour mettre en pratique ce qui a été vu jusque-là, nous allons créer étape par étape une macro qui va colorer 10x10 cellules (en damier rouge et noir) à partir de la cellule sélectionnée, aperçu :
[image: apercu - boucles suite]
Voici le point de départ de l'exercice :
Sub exercice_boucles()

    Const NB_CASES As Integer = 10 'Nombre de cellules à colorer

    '...
    
End Sub
Pour commencer, ajoutez une boucle For qui va colorer en noir les cellules de la colonne A de 1 à 10 (10 étant la constante NB_CASES), aperçu :
[image: 1 - boucles suite]
Sub exercice_boucles()

    Const NB_CASES As Integer = 10 'Nombre de cellules à colorer

    For l = 1 To NB_CASES 'l => n° ligne
    
        Cells(l, 1).Interior.Color = RGB(0, 0, 0) 'Noir

    Next
    
End Sub
La prochaine étape est la coloration en rouge d'une cellule sur 2 grâce à une instruction If (en fonction des n° de ligne pairs/impairs), aperçu :
[image: 2 - boucles suite]
.

Sub exercice_boucles()

    Const NB_CASES As Integer = 10 'Nombre de cellules à colorer

    For l = 1 To NB_CASES 'l => n° ligne
    
        If l Mod 2 = 0 Then 'Mod => est le reste d'une division
            Cells(l, 1).Interior.Color = RGB(200, 0, 0) 'Rouge
        Else
            Cells(l, 1).Interior.Color = RGB(0, 0, 0) 'Noir
        End If

    Next
    
End Sub
La condition If l Mod 2 = 0 signifie : si le reste de la division de l par 2 est égal à 0 ...
Seuls les n° de ligne pairs ont un reste égal à 0 lorsqu'ils sont divisés par 2.
Créez maintenant une boucle qui va exécuter la première boucle sur 10 colonnes, aperçu :
[image: 3 - boucles suite]
.
Sub exercice_boucles()

    Const NB_CASES As Integer = 10 'Damier de 10x10 cellules

    For l = 1 To NB_CASES 'l => n° ligne
    
        For c = 1 To NB_CASES 'c => n° colonne
        
            If l Mod 2 = 0 Then
                Cells(l, c).Interior.Color = RGB(200, 0, 0) 'Rouge
            Else
                Cells(l, c).Interior.Color = RGB(0, 0, 0) 'Noir
            End If
            
        Next
    Next
    
End Sub
La seconde boucle est donc imbriquée dans la première.
Pour obtenir ce résultat ...
[image: 4 - boucles suite]
Remplacez :
If l Mod 2 = 0 Then
Par :
If (l + c) Mod 2 = 0 Then
Il ne reste plus qu'à modifier le code pour créer le damier à partir de la cellule active (à la place de A1), aperçu :
[image: 5 - boucles suite]
Sub exercice_boucles()

    Const NB_CASES As Integer = 10 'Damier de 10x10 cellules
    Dim lig As Integer, col As Integer ' => ajout de 2 variables
    
    'Décalage (lignes) à partir de la première cellule = n° de ligne de la cellule active - 1
    lig = ActiveCell.Row - 1
    'Décalage (colonnes) à partir de la première cellule = n° de colonne de la cellule active - 1
    col = ActiveCell.Column - 1
    
    For l = 1 To NB_CASES 'N° ligne
    
        For c = 1 To NB_CASES 'N° colonne
        
            If (l + c) Mod 2 = 0 Then
                'Cells(n° de ligne + décalage lignes, n° de colonne + décalage colonnes)...
                Cells(l + lig, c + col).Interior.Color = RGB(200, 0, 0) 'Rouge
            Else
                Cells(l + lig, c + col).Interior.Color = RGB(0, 0, 0) 'Noir
            End If
            
        Next
    Next
    
End Sub
Télécharger ce cours au format les procédures et fonctions
Public - Private
Pour le moment, toutes les procédures créées sont de type Public, elles sont accessibles depuis tous les modules.
Sub exemple()
'Identique à :
Public Sub exemple()
Pour rendre une procédure inaccessible hors du module, ajoutez Private :
Private Sub exemple()


Lancer une procédure depuis une procédure
Pour exécuter une procédure depuis une autre procédure, entrez simplement son nom.
Un exemple simple :
Private Sub avertissement()
    MsgBox "Attention !!!"
End Sub

Sub macro_test()
    If Range("A1") = "" Then
        avertissement ' <= exécute la procédure "avertissement"
    End If
    'etc ...
End Sub
Ici, lorsque "macro_test" est exécutée et que A1 vaut "", la procédure "avertissement" est exécutée.
Les arguments
Les arguments permettent d'utiliser des valeurs d'une procédure dans une sous-procédure (car rappelez-vous que par défaut les variables ne sont pas accessibles depuis les autres procédures).
Private Sub avertissement(texte As String)
    MsgBox "Attention : " & texte & " !"
End Sub

Sub macro_test()
    If Range("A1") = "" Then
        avertissement "cellule vide"
    ElseIf Not IsNumeric(Range("A1")) Then
        avertissement "valeur non numérique"
    End If
End Sub
[image: test - procedures fonctions]
Un argument a été ajouté à la procédure "avertissement", il s'agit de la variable "texte" de type "String" :
Private Sub avertissement(texte As String)
Cette procédure nécessite un argument, il faudra donc placer une valeur après "avertissement" pour l'exécuter :
avertissement "cellule vide"
En cas d'arguments multiples, ceux-ci doivent être séparés par des virgules.
Les arguments optionnels
Par défaut, si une procédure requiert des arguments, ceux-ci sont obligatoires pour exécuter la procédure.
Des arguments optionnels peuvent être ajoutés après les arguments obligatoires avec Optional, par exemple :
Private Sub boite_de_dialogue(nom As String, Optional prenom, Optional age)
Cette procédure peut alors être lancée avec ou sans arguments optionnels, comme ceci :
'Exemple 1 : on affiche le nom :
boite_de_dialogue nom1
    
'Exemple 2 : on affiche le nom et le prénom :
boite_de_dialogue nom1, prenom1
    
'Exemple 3 : on affiche le nom et l'âge :
boite_de_dialogue nom1, , age1
    
'Exemple 4 : on affiche le nom, le prénom et l'âge :
boite_de_dialogue nom1, prenom1, age1
Les arguments doivent être indiqués dans l'ordre.
Pour vérifier si un argument optionnel est présent ou non, nous utiliserons la fonction IsMissing. Cette fonction n'étant compatible qu'avec certains types de variables (dont Variant), le type des arguments optionnels n'a pas été déclaré (type non déclaré = Variant).
Voici un exemple avec les 2 portions de code ci-dessus :
Sub macro_test()

    Dim nom1 As String, prenom1 As String, age1 As Integer
    
    nom1 = Range("A1")
    prenom1 = Range("B1")
    age1 = Range("C1")

    'Exemple 1 : on affiche le nom :
    boite_de_dialogue nom1
    
    'Exemple 2 : on affiche le nom et le prénom :
    boite_de_dialogue nom1, prenom1
    
    'Exemple 3 : on affiche le nom et l'âge :
    boite_de_dialogue nom1, , age1
    
    'Exemple 4 : on affiche le nom, le prénom et l'âge :
    boite_de_dialogue nom1, prenom1, age1

End Sub

Private Sub boite_de_dialogue(nom As String, Optional prenom, Optional age)
    
    If IsMissing(age) Then 'Si la variable age est absente ...
        
        If IsMissing(prenom) Then 'Si la variable prenom est absente, on n'affiche que le nom
            MsgBox nom
        Else 'Sinon, on affiche le nom et le prénom
            MsgBox nom & " " & prenom
        End If
        
    Else 'Si la variable age est présente ...

        If IsMissing(prenom) Then 'Si la variable prenom est absente, on affiche le nom et l'âge
            MsgBox nom & ", " & age & " ans"
        Else 'Sinon on affiche le nom, le prénom et l'âge
            MsgBox nom & " " & prenom & ", " & age & " ans"
        End If
    
    End If
       
End Sub
Aperçu (exemple 1) :
[image: test2 - procedures fonctions]
ByRef - ByVal
Par défaut, les arguments sont de type ByRef ce qui signifie que, si une variable est passée en argument, c'est sa référence qui est transmise. Autrement dit, si la variable est modifiée dans la sous-procédure, elle le sera également dans la procédure d'appel.
Par exemple :
Sub macro_test()
    Dim nombre As Integer
    nombre = 30
    
    calcul_carre nombre
    
    MsgBox nombre
End Sub

Private Sub calcul_carre(ByRef valeur As Integer) 'Il n'est pas nécessaire de préciser ByRef (puisque par défaut)
    valeur = valeur * valeur
End Sub
Pour mieux comprendre, voici ce qui se passe lorsque la macro est lancée :
nombre = 30
'La valeur initiale de la variable "nombre" est 30

calcul_carre nombre
'La sous procédure est lancée avec la variable "nombre" en argument

Private Sub calcul_carre(ByRef valeur As Integer)
'La variable "valeur" est en quelque sorte un raccourci vers la variable "nombre", par conséquent, si la variable "valeur" est modifiée, cela modifie la variable "nombre" (il n'est pas nécessaire de les nommer de façon identique)

valeur = valeur * valeur
'La valeur de la variable "valeur" est modifiée (donc la variable "nombre" est modifiée)

End Sub
'Fin de la sous-procédure

MsgBox nombre
'La variable "nombre" a été modifiée, 900 est alors affiché dans la boîte de dialogue
La seconde possibilité consiste à utiliser ByVal.
Contrairement à ByRef qui transmet la référence (raccourci), ByVal transmet la valeur, ce qui signifie que la variable passée en argument ne subit aucune modification.
Voici ce qui se passe avec le code précédent et ByVal :
nombre = 30
'La valeur initiale de la variable "nombre" est 30

calcul_carre nombre
'La sous procédure est lancée avec la variable "nombre" en argument

Private Sub calcul_carre(ByVal valeur As Integer)
'La variable "valeur" copie la valeur de la variable "nombre" (les 2 variables ne sont pas liées)

valeur = valeur * valeur
'La valeur de la variable "valeur" est modifiée

End Sub
'Fin de la sous-procédure (dans cet exemple, la sous-procédure n'aura servi à rien)

MsgBox nombre
'La variable "nombre" n'a pas été modifiée, 30 est donc affiché dans la boîte de dialogue
Ce qu'il faut retenir : utiliser ByVal lorsque la variable ne doit pas être modifiée ...
Les fonctions
La principale différence entre une procédure Sub et Function est la valeur retournée par la fonction.
En voici un exemple simple :
Function carre(nombre)
    carre = nombre ^ 2 'La fonction "carre" renvoie la valeur de "carre"
End Function

Sub macro_test()
    Dim resultat As Double
    resultat = carre(9.876) 'La variable resultat reçoit la valeur retournée par la fonction
    MsgBox resultat 'Affiche le résultat (ici, le carré de 9.876)
End Sub
Les fonctions peuvent également être utilisées sur la feuille comme n'importe quelle fonction Excel.
Par exemple, pour obtenir le carré de la valeur de A1 :
[image: carre - procedures fonctions]
les boîtes de dialogue
MsgBox
Pour le moment, nous n'avons utilisé la boîte de dialogue MsgBox que pour afficher une information :
Sub effacer_B2()
    Range("B2").ClearContents
    MsgBox "Le contenu de B2 a été effacé !"
End Sub
[image: 1 - boites de dialogue]
Nous allons maintenant créer une boîte de dialogue qui va nous demander de confirmer la suppression avant d'effectuer les deux instructions. Voici les 3 arguments que nous allons renseigner :
MsgBox([TEXTE], [BOUTONS], [TITRE])
· Texte : texte de la boîte de dialogue
· Boutons : choix des boutons (oui, non, annuler, etc.) + autres options
· Titre : titre de la boîte de dialogue
Sub effacer_B2()
    If MsgBox("Etes-vous certain de vouloir supprimer le contenu de B2 ?", vbYesNo, "Demande de confirmation") = vbYes Then
        Range("B2").ClearContents
        MsgBox "Le contenu de B2 a été effacé !"
    End If
End Sub
Aperçu :
[image: confirmation - boites de dialogue]
vbYesNo indique que les boutons de la boîte de dialogue sont "Oui" et "Non", vbYes correspond au bouton "Oui" :
If MsgBox("Texte", vbYesNo, "Titre") = vbYes Then 'Si le bouton Oui est cliqué ...
	Constante
	Valeur
	Description

	vbOKOnly
	0
	[image: ok - boites de dialogue]

	vbOKCancel
	1
	[image: ok - boites de dialogue][image: annuler - boites de dialogue]

	vbAbortRetryIgnore
	2
	[image: abandonner - boites de dialogue][image: recommencer - boites de dialogue][image: ignorer - boites de dialogue]

	vbYesNoCancel
	3
	[image: oui - boites de dialogue][image: non - boites de dialogue][image: annuler - boites de dialogue]

	vbYesNo
	4
	[image: oui - boites de dialogue][image: non - boites de dialogue]

	vbRetryCancel
	5
	[image: recommencer - boites de dialogue][image: annuler - boites de dialogue]



	vbCritical
	16
	[image: critique - boites de dialogue]

	vbQuestion
	32
	[image: question - boites de dialogue]

	vbExclamation
	48
	[image: exclamation - boites de dialogue]

	vbInformation
	64
	[image: information - boites de dialogue]



	vbDefaultButton1
	0
	Bouton par défaut : Bouton 1

	vbDefaultButton2
	256
	Bouton par défaut : Bouton 2

	vbDefaultButton3
	512
	Bouton par défaut : Bouton 3



	vbApplicationModal
	0
	Force l'utilisateur à répondre avant de poursuivre avec Excel

	vbSystemModal
	4096
	Force l'utilisateur à répondre avant de poursuivre avec d'autres applications (boîte de dialogue au premier plan)


Les valeurs à 0 sont celles par défaut.
Le deuxième argument de MsgBox peut prendre jusqu'à une valeur de chaque tableau.
Par exemple, pour une boîte de dialogue avec "Oui, Non, Annuler" + icône exclamation + bouton 2 par défaut :
MsgBox("Texte", vbYesNoCancel + vbExclamation + vbDefaultButton2, "Titre")
Aperçu :
[image: 2 - boites de dialogue]
Les constantes peuvent être remplacées par leur valeur respective, ces 3 lignes sont identiques :
MsgBox("Texte", vbYesNoCancel + vbExclamation + vbDefaultButton2, "Titre")
MsgBox("Texte", 3 + 48 + 256, "Titre")
MsgBox("Texte", 307, "Titre")
Les valeurs renvoyées par MsgBox
	Constante
	Valeur
	Bouton correspondant à la valeur

	vbOK
	1
	[image: ok - boites de dialogue]

	vbCancel
	2
	[image: annuler - boites de dialogue]

	vbAbort
	3
	[image: abandonner - boites de dialogue]

	vbRetry
	4
	[image: recommencer - boites de dialogue]

	vbIgnore
	5
	[image: ignorer - boites de dialogue]

	vbYes
	6
	[image: oui - boites de dialogue]

	vbNo
	7
	[image: non - boites de dialogue]


Voici l'exemple d'une MsgBox qui apparaît en boucle tant que le bouton Oui n'est pas cliqué :
Sub petite_blague()
    Do
        If MsgBox("Aimez-vous le site Excel-Pratique ?", vbYesNo, "Sondage") = vbYes Then
            Exit Do ' => Si réponse = Oui on sort de la boucle
        End If
    Loop While 1 = 1 ' => Boucle infinie
    MsgBox ";-)"
End Sub
Saut de ligne dans une MsgBox
Pour aller à la ligne, vous pouvez insérer le caractère "saut de ligne" avec la fonction Chr et le n°10 correspondant à ce caractère, exemple :
MsgBox "Exemple 1" & Chr(10) & "Exemple 2" & Chr(10) & Chr(10) & "Exemple 3"
Aperçu :
[image: 10 - boites de dialogue]
InputBox
L'InputBox demande à l'utilisateur d'entrer une valeur depuis une boîte de dialogue, exemple :
Sub exemple()
    Dim resultat As String
    
    resultat = InputBox("Texte ?", "Titre") 'La variable reçoit la valeur entrée dans l'InputBox
    
    If resultat <> "" Then 'Si la valeur est différente de "" on affiche le résultat
        MsgBox resultat
    End If
End Sub
Aperçu :
[image: input - boites de dialogue]
Il est possible d'indiquer une valeur par défaut en troisième argument :
InputBox("Texte ?", "Titre", "Valeur par défaut")
Aperçu :
[image: input2 - boites de dialogue]
Télécharger ce cours au format PDFF

les événements Workbook
Nous pouvons exécuter du code lors de certains événements du classeur (ouverture, fermeture, etc).


Workbook_Open (à l'ouverture)
Pour exécuter des instructions à l'ouverture du classeur, rendez-vous dans ThisWorkbook et sélectionnez Workbook :
[image: workbook - evenements classeur]
L'événement Workbook_Open est ajouté par défaut, il agit à l'ouverture du classeur :
Private Sub Workbook_Open()

End Sub
Par exemple, en ajoutant l'instruction suivante, une boîte de dialogue sera affichée à l'ouverture du classeur :
Private Sub Workbook_Open()
    MsgBox "Message de bienvenue"
End Sub
Workbook_BeforeClose (avant fermeture)
Pour éxécuter des instructions juste avant la fermeture du classeur, choisissez BeforeClose :
[image: close - evenements classeur]
Private Sub Workbook_BeforeClose(Cancel As Boolean)

End Sub
La fermeture du classeur peut être annulée en attribuant la valeur True à la variable "Cancel".
Voici un exemple où l'utilisateur doit confirmer la fermeture du classeur :
Private Sub Workbook_BeforeClose(Cancel As Boolean)
    'Si l'utilisateur répond NON, la variable Cancel vaudra TRUE (ce qui annulera la fermeture)
    If MsgBox("Etes-vous certain de vouloir fermer ce classeur ?", 36, "Confirmation") = vbNo Then
        Cancel = True
    End If
End Sub
Workbook_BeforeSave (avant enregistrement)
Cet événement se déclenche juste avant l'enregistrement :
Private Sub Workbook_BeforeSave(ByVal SaveAsUI As Boolean, Cancel As Boolean)

End Sub
L'enregistrement peut être annulé en attribuant la valeur True à la variable "Cancel".
Workbook_BeforePrint (avant impression)
Cet événement se déclenche juste avant l'impression :
Private Sub Workbook_BeforePrint(Cancel As Boolean)

End Sub
L'impression peut être annulée en attribuant la valeur True à la variable "Cancel".
Workbook_AfterSave (après enregistrement)
Cet événement se déclenche juste après l'enregistrement :
Private Sub Workbook_AfterSave(ByVal Success As Boolean)

End Sub
Workbook_SheetActivate (à l'activation d'une feuille)
Cet événement se déclenche à chaque changement de feuille :
Private Sub Workbook_SheetActivate(ByVal Sh As Object)

End Sub
Par exemple, affichage du nom de la feuille dans une boîte de dialogue :
Private Sub Workbook_SheetActivate(ByVal Sh As Object)
    MsgBox "Nom de la feuille : " & Sh.Name
End Sub
Workbook_SheetBeforeDoubleClick (avant double-clic)
Cet événement se déclenche juste avant un double-clic sur une cellule :
Private Sub Workbook_SheetBeforeDoubleClick(ByVal Sh As Object, ByVal Target As Range, Cancel As Boolean)

End Sub
Par exemple, coloration d'une cellule double-cliquée en fonction de la feuille :
Private Sub Workbook_SheetBeforeDoubleClick(ByVal Sh As Object, ByVal Target As Range, Cancel As Boolean)
    If Sh.Name = "Feuil1" Then
        Target.Interior.Color = RGB(255, 108, 0) 'Couleur orange
    Else
        Target.Interior.Color = RGB(136, 255, 0) 'Couleur verte
    End If
End Sub
Workbook_SheetBeforeRightClick (avant clic droit)
Cet événement se déclenche juste avant un clic droit sur une cellule :
Private Sub Workbook_SheetBeforeRightClick(ByVal Sh As Object, ByVal Target As Range, Cancel As Boolean)

End Sub
Workbook_SheetChange (à chaque modification de cellule)
Cet événement se déclenche à chaque modification du contenu d'une cellule :
Private Sub Workbook_SheetChange(ByVal Sh As Object, ByVal Target As Range)

End Sub
Workbook_SheetCalculate (à chaque recalcul)
Cet événement se déclenche à chaque recalcul de données d'une feuille :
Private Sub Workbook_SheetCalculate(ByVal Sh As Object)

End Sub
Workbook_SheetSelectionChange (à chaque changement de sélection)
Cet événement se déclenche à chaque changement de sélection sur une feuille de calcul :
Private Sub Workbook_SheetSelectionChange(ByVal Sh As Object, ByVal Target As Range)

End Sub
Par exemple, coloration de la sélection si A1 est vide :
Private Sub Workbook_SheetSelectionChange(ByVal Sh As Object, ByVal Target As Range)
    If Range("A1") = "" Then
        Target.Interior.Color = RGB(124, 255, 255) 'Bleu clair
    End If
End Sub
Workbook_NewSheet (à l'insertion d'une feuille)
Cet événement se déclenche à chaque insertion d'une nouvelle feuille :
Private Sub Workbook_NewSheet(ByVal Sh As Object)

End Sub
Workbook_SheetFollowHyperlink (au clic sur un lien)
Cet événement se déclenche lors d'un clic sur un lien hypertexte :
Private Sub Workbook_SheetFollowHyperlink(ByVal Sh As Object, ByVal Target As Hyperlink)

End Sub
les événements Worksheet
A la page précédente les événements concernaient le classeur entier. Sur celle-ci, nous allons nous focaliser sur les événements liés à une feuille.


Worksheet_SelectionChange (au changement de sélection)
Pour exécuter des instructions en fonction d'un événement pour une feuille en particulier, sélectionnez la feuille dans l'éditeur, puis Worksheet :
[image: worksheet - evenements feuille]
L'événement SelectionChange est ajouté par défaut, il agit lors d'un changement de sélection :
Private Sub Worksheet_SelectionChange(ByVal Target As Range)

End Sub
Par exemple, voici un code qui colore la ou les cellules sélectionnées et qui supprime automatiquement la coloration de la dernière sélection lors d'un changement de sélection :
Private Sub Worksheet_SelectionChange(ByVal Target As Range)
    Static selection_precedente As String

    If selection_precedente <> "" Then
        'Suppression de la couleur de fond de la sélection précédente :
        Range(selection_precedente).Interior.ColorIndex = xlColorIndexNone
    End If

    'Coloration de la sélection actuelle :
    Target.Interior.Color = RGB(181, 244, 0)

    'Enregistrement de l'adresse de la sélection actuelle :
    selection_precedente = Target.Address
End Sub
Worksheet_Activate (à l'activation de la feuille)
Cet événement se déclenche lorsque la feuille est activée :
Private Sub Worksheet_Activate()

End Sub
Worksheet_Deactivate (à la sortie de la feuille)
Cet événement se déclenche lorsqu'une autre feuille est activée :
Private Sub Worksheet_Deactivate()

End Sub
Worksheet_BeforeDoubleClick (au double-clic)
Cet événement se déclenche lors d'un double-clic sur une cellule de la feuille :
Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, Cancel As Boolean)

End Sub
Worksheet_BeforeRightClick (au clic droit)
Cet événement se déclenche lors d'un clic droit sur la feuille :
Private Sub Worksheet_BeforeRightClick(ByVal Target As Range, Cancel As Boolean)

End Sub
Worksheet_Calculate (au recalcul de la feuille)
Cet événement se déclenche lorsque les données de la feuille sont recalculées :
Private Sub Worksheet_Calculate()

End Sub
Worksheet_Change (à chaque modification de cellule)
Cet événement se déclenche lors de modifications du contenu des cellules de la feuille :
Private Sub Worksheet_Change(ByVal Target As Range)

End Sub
Worksheet_FollowHyperlink (au clic sur un lien)
Cet événement se déclenche lors d'un clic sur un lien hypertexte :
Private Sub Worksheet_FollowHyperlink(ByVal Target As Hyperlink)

End Sub

Désactiver temporairement tous les événements
Pour exécuter du code sans déclencher d'événements, placez-le entre ces deux lignes :
Application.EnableEvents = False ' => désactive les événements
'Instructions
Application.EnableEvents = True ' => réactive les événements
UserForm
Pour ajouter un UserForm, procédez de la même manière que pour un nouveau module :
[image: usf - userform]

[image: usf2 - userform]
Si la fenêtre des propriétés n'est pas présente, affichez-la et commencez par modifier le nom de l'UserForm (pour mieux s'y retrouver par la suite) :
[image: evenements - userform][image: height - userform][image: proprietes - userform]
Tout comme le classeur ou les feuilles, l'UserForm a ses propres événements. Pour ajouter des événements, double-cliquez sur la fenêtre de l'UserForm :

Pour prendre un exemple, nous allons créer deux événements. Le premier pour définir les dimensions initiales de l'UserForm et le second pour augmenter ses dimensions de 50 pixels au clic.
L'événement UserForm_Initialize se déclenche au lancement de l'UserForm :

La propriété Height est la hauteur, Width la largeur :
Private Sub UserForm_Initialize()
    Mon_userform.Height = 100
    Mon_userform.Width = 100
End Sub
Pour simplifier le code, nous pouvons remplacer le nom de l'UserForm par Me (puisque ce code est placé dans l'UserForm sur lequel on souhaite agir) :


Private Sub UserForm_Initialize()
    Me.Height = 100
    Me.Width = 100
End Sub
Le second événement est déclenché au clic sur l'UserForm :
[image: fenetre - userform]Private Sub UserForm_Initialize()
    Me.Height = 100
    Me.Width = 100
End Sub

Private Sub UserForm_Click()
    Me.Height = Me.Height + 50
    Me.Width = Me.Width + 50
End Sub



Lancer un UserForm
Pour lancer un UserForm depuis une procédure, utilisez Show :
Sub lancer_userform()
    Mon_userform.Show
End Sub





les contrôles
Les contrôles ont également toute une panoplie de propriétés et d'événements qui diffèrent d'un contrôle à l'autre, nous ne verrons ici que quelques unes des nombreuses possibilités liées à ces contrôles.
[image: num - controles]Commencez par ajouter les 3 contrôles suivants : un intitulé (Label), une zone de texte (TextBox) et un bouton (CommandButton) :
[image: evenements - controles][image: label - controles]
Modifiez le nom et les propriétés des contrôles (dont la propriété Caption pour le texte) pour obtenir ceci :

Pour le moment, lorsque l'on rentre un numéro et que l'on valide, il ne se passe rien.
Pour y remédier, nous allons commencer par ajouter un événement pour entrer la valeur de la zone de texte dans la cellule A1 et fermer l'UserForm.
En double-cliquant sur un contrôle, vous aurez accès à ceci :

La liste déroulante contient les différents contrôles ainsi que l'UserForm.
Choisissez le bouton et l'événement Click :
Private Sub CommandButton_valider_Click()

    Range("A1") = TextBox_numero.Value
    'TextBox_numero est le nom de la zone de texte
    'Value est la propriété contenant la valeur de la zone de texte
    
    Unload Me
    'Unload ferme l'UserForm
    'Le nom de l'UserForm a été remplacé par Me (puisque ce code est placé dans l'UserForm à fermer)
End Sub
La valeur est alors enregistrée dans A1 avant la fermeture de l'UserForm.
Ajoutez un second Label et modifiez les propriétés suivantes : Caption, Forecolor (couleur rouge) et Visible (False, pour masquer le contrôle par défaut) :
[image: erreur - controles]
Nous allons maintenant ajouter un événement qui s'active au changement de valeur de la zone de texte et qui va afficher le message d'erreur si la valeur n'est pas numérique.
Private Sub TextBox_numero_Change()
    If IsNumeric(TextBox_numero.Value) Then 'SI valeur numérique ...
        Label_erreur.Visible = False 'Label masqué
    Else 'SINON ...
        Label_erreur.Visible = True 'Label visible
    End If
End Sub
Le test de la valeur est effectué à chaque entrée de caractère ...
Il nous reste encore à empêcher la validation du formulaire si la valeur n'est pas numérique :
Private Sub CommandButton_valider_Click()
    If IsNumeric(TextBox_numero.Value) Then 'SI valeur numérique ...
        Range("A1") = TextBox_numero.Value 'Copie en A1
        Unload Me 'Fermeture
    Else 'SINON ...
        MsgBox "Valeur incorrecte"
    End If
End Sub
[image: erreur2 - controles]
Pour ne pas laisser vide la partie droite de l'UserForm lorsqu'il n'y a pas d'erreur, nous pouvons la réduire en modifiant la propriété Width de l'UserForm :
Private Sub TextBox_numero_Change()
    If IsNumeric(TextBox_numero.Value) Then 'SI valeur numérique ...
        Label_erreur.Visible = False 'Label masqué
        Me.Width = 156 'Largeur de l'UserForm
    Else 'SINON ...
        Label_erreur.Visible = True 'Label visible
        Me.Width = 244 'Largeur de l'UserForm
    End If
End Sub
Le fichier : userform1.xls
Aperçu :
[image: reduc - controles] [image: reduc2 - controles]



Les cases à cocher (CheckBox)
Voici un exemple d'utilisation de cases à cocher :
[image: cc - controles]
Si une case est cochée/décochée la valeur de la cellule correspondante sera modifiée, avec l'événement Click :
Private Sub CheckBox1_Click() 'N°1
    If CheckBox1.Value = True Then 'Si coché ...
        Range("A2") = "Coché"
    Else 'Si non coché ...
        Range("A2") = "Non coché"
    End If
End Sub

Private Sub CheckBox2_Click() 'N°2
    If CheckBox2.Value = True Then 'Si coché ...
        Range("B2") = "Coché"
    Else 'Si non coché ...
        Range("B2") = "Non coché"
    End If
End Sub

Private Sub CheckBox3_Click() 'N°3
    If CheckBox3.Value = True Then 'Si coché ...
        Range("C2") = "Coché"
    Else 'Si non coché ...
        Range("C2") = "Non coché"
    End If
End Sub
[image: cc2 - controles]
Dans cet exemple, les cases ne sont pas cochées au lancement de l'UserForm.
Pour cocher les cases dont la valeur de la cellule correspondante est "Coché", un test est effectué au lancement de l'UserForm avec l'événement UserForm_Initialize :
[image: bo - controles]Private Sub UserForm_Initialize() 'Cocher si "Coché"
    If Range("A2") = "Coché" Then
        CheckBox1.Value = True
    End If
    
    If Range("B2") = "Coché" Then
        CheckBox2.Value = True
    End If
    
    If Range("C2") = "Coché" Then
        CheckBox3.Value = True
    End If
End Sub
Le fichier : userform2.xls
Les boutons d'option (OptionButton)
Contrairement aux cases à cocher, l'utilisateur ne peut choisir qu'un seul bouton d'option par "groupe".
Pour créer un groupe, insérez d'abord un cadre (Frame) puis les boutons d'option :

Le fichier : userform3.xls
Une fois le formulaire validé, nous entrerons une donnée dans la cellule correspondant aux numéros de colonne et de ligne choisis.
Pour connaître le bouton qui a été coché, nous pourrions procéder de la même manière qu'avec le précédent exemple (cases à cocher) mais nous allons le faire à l'aide d'une boucle pour alléger le code.
Nous utiliserons la boucle For Each que nous n'avions pas encore abordé jusque-là. Elle permet d'exécuter des instructions pour chaque objet d'un "groupe d'objet" :
Private Sub CommandButton1_Click()
    Dim colonne As String, ligne As String
    
    'Boucle pour chaque contrôle de Frame_colonne
    For Each bouton_colonne In Frame_colonne.Controls
        'Si la valeur du contrôle = True (donc si coché) ...
        If bouton_colonne.Value Then
            'La variable "colonne" prend comme valeur le texte du bouton
            colonne = bouton_colonne.Caption
        End If
    Next
    
    'Boucle pour l'autre frame
    For Each bouton_ligne In Frame_ligne.Controls
        If bouton_ligne.Value Then
            ligne = bouton_ligne.Caption
        End If
    Next

    Range(colonne & ligne) = "Cellule choisie !"
    Unload Me
End Sub
Ce formulaire entre alors la valeur "Cellule choisie !" dans la cellule choisie (pour autant que le formulaire soit complet).
Pour éviter d'avoir ce bug, nous avons besoin de vérifier que l'utilisateur a bien coché les deux boutons d'option.
Pour cet exemple, tant que le formulaire sera incomplet, le bouton "Valider" restera désactivé. Ce n'est pas la solution la plus simple, mais cela vous démontrera l'intérêt d'utiliser des fonctions/procédures dans un UserForm.
Modifiez le texte ainsi que la propriété Enabled pour désactiver le bouton :
[image: bo3 - controles][image: bo2 - controles]
Ce qui nous donne :

Dans le code ci-dessus, nous avons utilisé 2 boucles For Each pour obtenir les valeurs des boutons cochés. Nous allons avoir besoin de ces mêmes valeurs dans les événements Click du bouton "Valider" ainsi que des 10 boutons d'option.
Pour éviter de copier ces boucles dans chaque événement, nous allons les appeler à l'aide d'une fonction.
En reprenant le dernier code et en le transformant, cela nous donne :
Private Function colonne()
'La fonction renvoie comme valeur le texte du bouton choisi (colonne)
    For Each bouton_colonne In Frame_colonne.Controls
        If bouton_colonne.Value Then
            colonne = bouton_colonne.Caption
        End If
    Next
End Function

Private Function ligne()
'La fonction renvoie comme valeur le texte du bouton choisi (ligne)
    For Each bouton_ligne In Frame_ligne.Controls
        If bouton_ligne.Value Then
            ligne = bouton_ligne.Caption
        End If
    Next
End Function

Private Sub CommandButton1_Click() 'Action quand clic sur "Valider le choix"
    Range(colonne & ligne) = "Cellule choisie !"
    'colonne et ligne sont les valeurs renvoyées par les fonctions
    Unload Me
End Sub
Il ne reste plus qu'à créer une procédure qui vérifie si les boutons sont bien cochés (en faisant appel aux deux fonctions) et qui active le bouton si c'est le cas.
Là encore, le test est effectué dans une procédure à part pour éviter de copier 10x le code dans les événements des boutons d'option :
Private Sub activer()
'Activation du bouton si la condition est vérifiée
    If colonne <> "" And ligne <> "" Then
    'colonne et ligne sont les valeurs renvoyées par les fonctions
        CommandButton1.Enabled = True
        CommandButton1.Caption = "Valider le choix"
    End If
End Sub

Private Sub OptionButton11_Click()
    activer 'Lance la procédure "activer"
End Sub
Private Sub OptionButton12_Click()
    activer
End Sub
Private Sub OptionButton13_Click()
    activer
End Sub
Private Sub OptionButton14_Click()
    activer
End Sub
Private Sub OptionButton15_Click()
    activer
End Sub
Private Sub OptionButton16_Click()
    activer
End Sub
Private Sub OptionButton17_Click()
    activer
End Sub
Private Sub OptionButton18_Click()
    activer
End Sub
Private Sub OptionButton19_Click()
    activer
End Sub
Private Sub OptionButton20_Click()
    activer
End Sub
[image: bo4 - controles]
Le fichier : userform3b.xls
 
Les barres de défilement (ScrollBar)
Les contrôles peuvent être utilisés en dehors des UserForms. Pour cet exemple, nous utiliserons les contrôles directement sur la feuille.
Notez tout d'abord que le "Mode Création" doit être activé pour modifier un contrôle placé sur une feuille (et désactivé pour utiliser le contrôle).
[image: mode - controles suite]
Pour les versions d'Excel inférieures à 2007 : bouton sur la barre "Boîte à outils Contrôles".
Avant de détailler cet exemple, en voici un aperçu :
[image: sb - controles suite]
L'objectif est de colorer et sélectionner une cellule grâce aux barres de défilement dans la zone définie de 30 lignes x 10 colonnes.
Les propriétés de la barre de défilement verticale :
· Min : 1
· Max : 30 (puisque 30 lignes)
· Value : position de la barre (entre 1 et 30 ici)
Même chose avec la barre horizontale avec un Max à 10 ...
Voici le code qui sera appliqué à chaque changement de valeur (Value) de la barre verticale :
'Couleur grise sur les cellules
Cells.Interior.Color = RGB(240, 240, 240)

'Couleur et sélection de la cellule
With Cells(ScrollBar_verticale.Value, ActiveCell.Column) 'Cellule par rapport à Value
    .Interior.Color = RGB(255, 220, 100) 'Couleur orange
    .Select 'Sélection de la cellule
End With
Ce code sera exécuté avec les événements Change et Scroll pour lancer les instructions quelle que soit la partie de la barre de défilement qui est cliquée.
Voici le code pour la barre verticale :
Private Sub barre_verticale()
    'Couleur grise sur les cellules
    Cells.Interior.Color = RGB(240, 240, 240)
    
    'Couleur et sélection de la cellule
    With Cells(ScrollBar_verticale.Value, ActiveCell.Column)
        .Interior.Color = RGB(255, 220, 100) 'Orange
        .Select 'Sélection de la cellule
    End With
End Sub

Private Sub ScrollBar_verticale_Change()
    barre_verticale
End Sub

Private Sub ScrollBar_verticale_Scroll()
    barre_verticale
End Sub
Et celui de la barre horizontale :
Private Sub barre_horizontale()
    'Couleur grise sur les cellules
    Cells.Interior.Color = RGB(240, 240, 240)

    'Couleur et sélection de la cellule
    With Cells(ActiveCell.Row, ScrollBar_horizontale.Value)
        .Interior.Color = RGB(255, 220, 100) 'Orange
        .Select 'Sélection de la cellule
    End With
End Sub

Private Sub ScrollBar_horizontale_Change()
    barre_horizontale
End Sub

Private Sub ScrollBar_horizontale_Scroll()
    barre_horizontale
End Sub
Le fichier : scrollbar.xls
La liste déroulante (ComboBox) et la zone de liste (ListBox)
Voici le point de départ de ce nouvel exemple :
[image: listes - controles suite]
Le fichier : userform4.xls
Au lancement de l'UserForm, nous voulons que les 4 pays soient chargés dans la liste déroulante (avec la méthode AddItem) :
Private Sub UserForm_Initialize()
    For i = 1 To 4 ' => pour lister les 4 pays
        ComboBox_Pays.AddItem Cells(1, i) 'Ajoute les valeurs des cellules A1 à D1 avec la boucle
    Next
End Sub
Au changement de valeur de la liste déroulante, nous voulons ajouter les villes correspondant au pays choisi avec une boucle semblable à celle ci-dessus.
Pour le faire, nous avons besoin du n° de colonne et du nombre de lignes.
La propriété ListIndex correspond au n° de la sélection dans la liste (contrairement à Value qui correspond à la valeur), à noter que ListIndex commence à 0.
Le n° de colonne est donc :
no_colonne = ComboBox_Pays.ListIndex + 1
Pour obtenir le nombre de lignes de la colonne du pays choisi, nous pouvons rechercher le n° de ligne de la dernière cellule d'un bloc de cellules non vides :
nb_lignes = Cells(1, no_colonne).End(xlDown).Row
Grâce à ces informations, il est désormais possible de créer la boucle pour ajouter les villes dans la zone de liste :
[image: listes2 - controles suite]Private Sub ComboBox_Pays_Change()
    'Zone de liste vidée (sinon les villes sont ajoutées à la suite)
    ListBox_Villes.Clear
    
    Dim no_colonne As Integer, nb_lignes As Integer
    
    'Numéro de la sélection (ListIndex commence à 0) :
    no_colonne = ComboBox_Pays.ListIndex + 1
    'Nombre de lignes de la colonne du pays choisi :
    nb_lignes = Cells(1, no_colonne).End(xlDown).Row

    For i = 2 To nb_lignes ' => pour lister les villes
        ListBox_Villes.AddItem Cells(i, no_colonne)
    Next
End Sub
Remarque : nous pourrions réduire le code ci-dessus, néanmoins cela rendrait sa lecture plus difficile :
Private Sub ComboBox_Pays_Change()
    ListBox_Villes.Clear
    For i = 2 To Cells(1, ComboBox_Pays.ListIndex + 1).End(xlDown).Row
        ListBox_Villes.AddItem Cells(i, ComboBox_Pays.ListIndex + 1)
    Next
End Sub
La ville choisie est ensuite entrée dans la zone de texte :
Private Sub ListBox_Villes_Click()
    TextBox_Choix.Value = ListBox_Villes.Value
End Sub

Le fichier : userform4b.xls
Aller plus loin
N'oubliez pas qu'avec l'aide Excel vous pouvez obtenir des informations rapidement sur les différentes propriétés et événements des contrôles (entre autres).
En voici un aperçu avec la recherche du contrôle Label :
[image: aide - controles suite]
L'aide Excel est très riche en informations, n'hésitez pas à l'utiliser en cas de besoin ...
Cette aide très pratique a malheureusement été supprimée des dernières versions d'Excel.
les contrôles (exercice)
Pour mettre en pratique l'utilisation des contrôles, rien de tel qu'un petit exercice ...
Voici le point de départ de l'exercice :
Le fichier : controles_exercice.xls
[image: 1 - controles exercice][image: depart - controles exercice]
.
Voici une solution pour réaliser cet exercice
La première action effectuée a été d'augmenter la propriété Zoom de l'UserForm à 120 pour plus de confort à l'utilisation du formulaire :

Les tests des boutons d'option ont déjà été vus à la première page des contrôles, c'est pour cela qu'une solution plus simple a été utilisée ici.
Le bouton "Mme" a été choisi par défaut (propriété Value : True), cela implique qu'il ne sera pas nécessaire de vérifier si le choix de la civilité a été effectué.
Le bouton "Fermer"
Private Sub CommandButton_Fermer_Click()
    Unload Me
End Sub
Le contenu de la liste déroulante
Private Sub UserForm_Initialize() 'Chargement de la liste au lancement de l'UserForm
    For i = 1 To 231 'Liste des 231 pays de la feuille "Pays"
        ComboBox_Pays.AddItem Sheets("Pays").Cells(i, 1)
    Next
End Sub
Vérification des contrôles
Une solution simple consiste à afficher une boîte de dialogue si l'un des contrôles est vide :
Private Sub CommandButton_Ajouter_Click()
    If TextBox_Nom.Value = "" Or TextBox_Prenom.Value = "" Or TextBox_Adresse.Value = "" Or TextBox_Lieu.Value = "" Or ComboBox_Pays.Value = "" Then
        MsgBox "Formulaire incomplet"
    Else
        'Instructions pour insérer le contact ici ...
    End If
End Sub
Mais pour compliquer un peu les choses, chaque contrôle sera testé individuellement, et si l'un d'eux n'est pas rempli, son intitulé (Label) sera coloré en rouge :
Private Sub CommandButton_Ajouter_Click()
    'Coloration des Labels en noir
    Label_Nom.ForeColor = RGB(0, 0, 0)
    Label_Prenom.ForeColor = RGB(0, 0, 0)
    Label_Adresse.ForeColor = RGB(0, 0, 0)
    Label_Lieu.ForeColor = RGB(0, 0, 0)
    Label_Pays.ForeColor = RGB(0, 0, 0)

    'Contrôles de contenu
    If TextBox_Nom.Value = "" Then 'SI pas de "nom" ...
        Label_Nom.ForeColor = RGB(255, 0, 0) 'Label "nom" en rouge
    ElseIf TextBox_Prenom.Value = "" Then
        Label_Prenom.ForeColor = RGB(255, 0, 0)
    ElseIf TextBox_Adresse.Value = "" Then
        Label_Adresse.ForeColor = RGB(255, 0, 0)
    ElseIf TextBox_Lieu.Value = "" Then
        Label_Lieu.ForeColor = RGB(255, 0, 0)
    ElseIf ComboBox_Pays.Value = "" Then
        Label_Pays.ForeColor = RGB(255, 0, 0)
    Else
        'Instructions pour insérer le contact ici ...
    End If
End Sub
[image: 2 - controles exercice]
Insertion des données
Le code suivant a été inséré à l'emplacement indiqué sur la code précédent (sous forme de commentaire) :
Dim no_ligne As Integer, civilite As String

'Choix de civilité
For Each bouton_civilite In Frame_Civilite.Controls
    If bouton_civilite.Value Then
        civilite = bouton_civilite.Caption 'Civilité choisie
    End If
Next

'no_ligne = N° de ligne de la dernière cellule non vide de la colonne +1
no_ligne = Range("A65536").End(xlUp).Row + 1

'Insertion des valeurs sur la feuille
Cells(no_ligne, 1) = civilite
Cells(no_ligne, 2) = TextBox_Nom.Value
Cells(no_ligne, 3) = TextBox_Prenom.Value
Cells(no_ligne, 4) = TextBox_Adresse.Value
Cells(no_ligne, 5) = TextBox_Lieu.Value
Cells(no_ligne, 6) = ComboBox_Pays.Value

'Après insertion, on remet les valeurs initiales
OptionButton1.Value = True
TextBox_Nom.Value = ""
TextBox_Prenom.Value = ""
TextBox_Adresse.Value = ""
TextBox_Lieu.Value = ""
ComboBox_Pays.ListIndex = -1
Vue d'ensemble
Pour terminer, voici le code complet ainsi que le fichier :
Private Sub CommandButton_Fermer_Click()
    Unload Me
End Sub

Private Sub UserForm_Initialize() 'Liste des 231 pays de la feuille "Pays"
    For i = 1 To 231
        ComboBox_Pays.AddItem Sheets("Pays").Cells(i, 1)
    Next
End Sub

Private Sub CommandButton_Ajouter_Click()
    'Coloration des Labels en noir
    Label_Nom.ForeColor = RGB(0, 0, 0)
    Label_Prenom.ForeColor = RGB(0, 0, 0)
    Label_Adresse.ForeColor = RGB(0, 0, 0)
    Label_Lieu.ForeColor = RGB(0, 0, 0)
    Label_Pays.ForeColor = RGB(0, 0, 0)

    'Contrôles de contenu
    If TextBox_Nom.Value = "" Then 'SI pas de "nom" ...
        Label_Nom.ForeColor = RGB(255, 0, 0) 'Label "nom" en rouge
    ElseIf TextBox_Prenom.Value = "" Then
        Label_Prenom.ForeColor = RGB(255, 0, 0)
    ElseIf TextBox_Adresse.Value = "" Then
        Label_Adresse.ForeColor = RGB(255, 0, 0)
    ElseIf TextBox_Lieu.Value = "" Then
        Label_Lieu.ForeColor = RGB(255, 0, 0)
    ElseIf ComboBox_Pays.Value = "" Then
        Label_Pays.ForeColor = RGB(255, 0, 0)
    Else
        'Si le formulaire est complet, on insère les valeurs sur la feuille
        Dim no_ligne As Integer, civilite As String
        
        'Choix de civilité
        For Each bouton_civilite In Frame_Civilite.Controls
            If bouton_civilite.Value Then
                civilite = bouton_civilite.Caption
            End If
        Next

        'no_ligne = N° de ligne de la dernière cellule non vide de la colonne +1
        no_ligne = Range("A65536").End(xlUp).Row + 1

        'Insertion des valeurs sur la feuille
        Cells(no_ligne, 1) = civilite
        Cells(no_ligne, 2) = TextBox_Nom.Value
        Cells(no_ligne, 3) = TextBox_Prenom.Value
        Cells(no_ligne, 4) = TextBox_Adresse.Value
        Cells(no_ligne, 5) = TextBox_Lieu.Value
        Cells(no_ligne, 6) = ComboBox_Pays.Value
        
        'Après insertion, on remet les valeurs initiales
        OptionButton1.Value = True
        TextBox_Nom.Value = ""
        TextBox_Prenom.Value = ""
        TextBox_Adresse.Value = ""
        TextBox_Lieu.Value = ""
        ComboBox_Pays.ListIndex = -1
    End If
End Sub
Télécharger ce cours au format PDF
utilisations des tableaux
Les tableaux sont des "variables" qui permettent de stocker une multitude de valeurs. Nous avons effleuré le sujet à la leçon sur les variables, nous allons maintenant l'approfondir ...
Intérêt des tableaux
Imaginez que dans une procédure vous ayez besoin de stocker 500 valeurs. S'il fallait créer 500 variables pour stocker toutes ces valeurs, cela deviendrait très compliqué, tandis qu'avec un tableau, le stockage et l'utilisation de ces valeurs seront grandement simplifiés.


Le second intérêt des tableaux est leur "rapidité". Parcourir un tableau de données demande beaucoup moins de temps que parcourir un "tableau" (composé de cellules) sur une feuille Excel ...
Rien de tel qu'un exemple pour mieux comprendre ...
Sur la première feuille ("BD") : une base de données de 5000 lignes sur 3 colonnes :
[image: avec opti - tableaux vba][image: bd - tableaux vba]


Dans cet exemple, l'utilisation d'un tableau a permis d'exécuter la procédure environ 128x plus rapidement et cette différence peut encore augmenter sensiblement lorsque l'on travaille avec plusieurs bases de données en même temps (voir un second exemple).
Vous l'avez compris, l'utilisation de tableaux peut vraiment faire la différence.
Cet exemple sera détaillé en fin de leçon ...
Déclaration d'un tableau
Voici quelques exemples de déclarations (si les 2 premiers ne sont pas clairs pour vous, relisez ceci) :
'Exemple de déclaration d'un tableau à 1 dimension
Dim tab1(4)

'Exemple de déclaration d'un tableau à 2 dimensions
Dim tab2(6, 1)

'Exemple de déclaration d'un tableau dynamique
Dim tab()
Si vous ne pouvez pas entrer de valeurs fixes (parce que cela dépend de la taille de la base de données par exemple), laissez les parenthèses vides.
Vous n'avez pas besoin de déclarer un type (string, long, etc.), dans bien des cas cela ralentirait votre procédure ...
Enregistrer des données dans un tableau
Commençons par enregistrer ces quelques données dans un tableau :
[image: enreg-1-dim - tableaux vba]
Nous voulons enregistrer ici 11 x 1 valeurs, il faudra donc créer un tableau à une dimension :
'Déclaration
Dim tab_exemple(10)
Rappelez-vous que la numérotation d'un tableau commence à 0 (c'est une norme en programmation, autant prendre de bonnes habitudes dès le début même s'il est possible de modifier cela en VBA).
Chaque élément du tableau reçoit ensuite sa valeur :
'Enregistrement des valeurs dans le tableau
tab_exemple(0) = Range("A2")
tab_exemple(1) = Range("A3")
tab_exemple(2) = Range("A4")
tab_exemple(3) = Range("A5")
tab_exemple(4) = Range("A6")
tab_exemple(5) = Range("A7")
tab_exemple(6) = Range("A8")
tab_exemple(7) = Range("A9")
tab_exemple(8) = Range("A10")
tab_exemple(9) = Range("A11")
tab_exemple(10) = Range("A12")
Vous pouvez utiliser, modifier chaque élément du tableau comme une variable.
En voici un exemple avec tab_exemple(8) :
Sub enreg_tab()
    'Déclaration
    Dim tab_exemple(10)
    
    'Enregistrement des valeurs dans le tableau
    tab_exemple(0) = Range("A2")
    tab_exemple(1) = Range("A3")
    tab_exemple(2) = Range("A4")
    tab_exemple(3) = Range("A5")
    tab_exemple(4) = Range("A6")
    tab_exemple(5) = Range("A7")
    tab_exemple(6) = Range("A8")
    tab_exemple(7) = Range("A9")
    tab_exemple(8) = Range("A10")
    tab_exemple(9) = Range("A11")
    tab_exemple(10) = Range("A12")
    
    'Test 1
    MsgBox tab_exemple(8) '=> renvoie : 04.02.2016
    
    'Modification de l'une des valeurs
    tab_exemple(8) = Year(tab_exemple(8))
    
    'Test 2
    MsgBox tab_exemple(8) '=> renvoie : 2016
End Sub
Pour enregistrer ce même tableau plus rapidement, une boucle For est toute indiquée :
'Déclaration
Dim tab_exemple(10)

'Enregistrement des valeurs dans le tableau
For i = 0 To 10
    tab_exemple(i) = Range("A" & i + 2)
Next
Le tableau à 2 dimensions
Pour enregistrer plusieurs colonnes de données, une dimension supplémentaire est nécessaire. En voici un exemple :
[image: enreg-2-dim - tableaux vba suite]
Enregistrement des données dans un tableau à 2 dimensions :
'Déclaration
Dim tab_exemple(10, 2) 'Tableau de 11 x 3 "cases"

'Enregistrement des valeurs dans le tableau
For i = 0 To 10
    tab_exemple(i, 0) = Range("A" & i + 2)
    tab_exemple(i, 1) = Range("B" & i + 2)
    tab_exemple(i, 2) = Range("C" & i + 2)
Next
Et quelques exemples de valeurs :
MsgBox tab_exemple(0, 0) '=> renvoie : 11.03.2026
MsgBox tab_exemple(0, 1) '=> renvoie : 24
MsgBox tab_exemple(9, 2) '=> renvoie : NON
MsgBox tab_exemple(10, 2) '=> renvoie : OUI
Le tableau dynamique
Imaginons que cette même base de données soit régulièrement mise à jour et que l'on ne puisse donc pas entrer de valeurs fixes à la déclaration ...
[image: enreg-dynamique - tableaux vba suite]
Pour connaître le n° de ligne de la dernière cellule d'un bloc de cellules non vides, autrement dit, la dernière ligne de notre base de données, utilisez cette formule :
derniere_ligne = Range("A1").End(xlDown).Row
Si vous entrez une variable lors de la déclaration, Excel ne l'acceptera pas.
Déclarez un tableau dynamique (parenthèses vides), puis définissez ses dimensions avec Redim :
Dim tab_exemple()
ReDim tab_exemple(derniere_ligne - 2, 2)
De cette manière vous enregistrerez automatiquement toutes les lignes de la base de données dans le tableau :
Sub enreg_tab()
    derniere_ligne = Range("A1").End(xlDown).Row 'Dernière ligne de la base de données

    Dim tab_exemple()
    ReDim tab_exemple(derniere_ligne - 2, 2)
    
    'Enregistrement des valeurs dans le tableau
    For i = 0 To derniere_ligne - 2
        tab_exemple(i, 0) = Range("A" & i + 2)
        tab_exemple(i, 1) = Range("B" & i + 2)
        tab_exemple(i, 2) = Range("C" & i + 2)
    Next
End Sub
Ubound
Dans l'exemple ci-dessus, le dernier n° de notre tableau était derniere_ligne - 2 :
For i = 0 To derniere_ligne - 2
Une autre solution pour connaître ce n° consiste à utiliser la fonction Ubound :
For i = 0 To UBound(tab_exemple)
Cette fonction renvoie le plus grand n° pour une dimension choisie (par défaut la première).
Quelques exemples pour mieux comprendre :
Sub enreg_tab()
    Dim tab_exemple(10, 2)
    
    MsgBox UBound(tab_exemple) '=> renvoie : 10
    MsgBox UBound(tab_exemple, 1) '=> renvoie : 10
    MsgBox UBound(tab_exemple, 2) '=> renvoie : 2
End Sub
Enregistrer une plage de cellules
Il est possible d'enregistrer une plage de cellules dans un tableau sans passer par une boucle.
'Déclaration
Dim tab_exemple(10, 2) 'Tableau de 11 x 3 "cases"

'Enregistrement des valeurs dans le tableau
For i = 0 To 10
    tab_exemple(i, 0) = Range("A" & i + 2)
    tab_exemple(i, 1) = Range("B" & i + 2)
    tab_exemple(i, 2) = Range("C" & i + 2)
Next
Le code ci-dessus peut être remplacé par :
'Déclaration
Dim tab_exemple()

'Enregistrement des valeurs dans le tableau
tab_exemple = Range("A2:C12").Value
Même si au premier abord cette seconde méthode semble séduisante, elle peut dans bien des cas vous faire perdre plus de temps que la première méthode ...
En enregistrant vos données dans le tableau de cette manière, le premier n° n'est pas 0 mais 1, cela peut être source de confusion ... De plus, si au cours du développement vous choisissez de n'enregistrer dans le tableau que les données répondant à certains critères (ou effectuer toute autre opération), vous aurez besoin de tout réécrire avec une boucle ...
Cette seconde méthode reste tout de même intéressante lorsque vous avez besoin d'enregistrer l'ensemble du contenu d'une grande base de données, car plus rapide qu'avec une boucle (gain d'environ 0.2 secondes pour 15'000 entrées).
Array
Vous aurez peut-être parfois besoin de créer un tableau contenant des données "fixes".
Une solution consiste à l'écrire ligne par ligne :
Dim fr(5)

fr(0) = "SI"
fr(1) = "RECHERCHEV"
fr(2) = "SOMME"
fr(3) = "NB"
fr(4) = "ESTNUM"
fr(5) = "STXT"
Heureusement, vous pouvez vous simplifier la tâche en utilisant Array :
fr = Array("SI", "RECHERCHEV", "SOMME", "NB", "ESTNUM", "STXT")
Voici un exemple d'utilisation de la fonction Replace (utile pour mieux comprendre l'exemple suivant) :
Sub remplacement()
    Dim chaine_a_traiter As String

    'Une chaîne pour cet exemple
     chaine_a_traiter = "Hello World !"
    
    'Remplacement de "World" par "toi" dans la chaîne de caractères
     chaine_a_traiter = Replace(chaine_a_traiter, "World", "toi")

    'La chaîne après remplacement
     MsgBox chaine_a_traiter '=> renvoie "Hello toi !"
End Sub
Maintenant si l'on veut remplacer une série de valeurs par une autre série, l'utilisation de tableaux (Array) est toute indiquée :
Sub traduction() 'Exemple simplifié de traduction FR-EN de formules
    Dim chaine_a_traiter As String

    'Une chaîne pour cet exemple
    chaine_a_traiter = "Formule à traduire : SOMME(SI(ESTNUM(A1:E1);A1:E1;0))"
    
    'Les 2 séries de valeurs
    fr = Array("SI", "RECHERCHEV", "SOMME", "NB", "ESTNUM", "STXT")
    en = Array("IF", "VLOOKUP", "SUM", "COUNT", "ISNUMBER", "MID")
    
    'Remplacement de "SI" par "IF", de "RECHERCHEV" par "VLOOKUP", etc.
    For i = 0 To UBound(fr)
        chaine_a_traiter = Replace(chaine_a_traiter, fr(i), en(i))
    Next

    'La chaîne après les remplacements
    MsgBox chaine_a_traiter '=> renvoie "Formule à traduire : SUM(IF(ISNUMBER(A1:E1);A1:E1;0))"
End Sub
Split
La fonction Split permet de convertir une chaîne de caractères en un tableau.
Pour convertir cette chaîne de caractères en tableau :
chaine = "SI/RECHERCHEV/SOMME/NB/ESTNUM/STXT"
Utilisez la fonction Split et définissez le séparateur :
fr = Split(chaine, "/")
Le tableau fr renverra les valeurs suivantes :
MsgBox fr(0) '=> renvoie : SI
MsgBox fr(1) '=> renvoie : RECHERCHEV
MsgBox fr(2) '=> renvoie : SOMME
MsgBox fr(3) '=> renvoie : NB
MsgBox fr(4) '=> renvoie : ESTNUM
MsgBox fr(5) '=> renvoie : STXT
Les 3 tableaux suivants renvoient également les mêmes valeurs :
fr = Array("SI", "RECHERCHEV", "SOMME", "NB", "ESTNUM", "STXT")
fr = Split("SI,RECHERCHEV,SOMME,NB,ESTNUM,STXT", ",")
fr = Split("SI RECHERCHEV SOMME NB ESTNUM STXT", " ")
L'exemple suivant renvoie la 3e valeur de la chaîne de caractères :
MsgBox Split("SI,RECHERCHEV,SOMME,NB,ESTNUM,STXT", ",")(2) '=> renvoie : SOMME
[image: join - tableaux vba suite][image: somme - tableaux vba suite]
La fonction à l'opposé de Split est Join.
Cette fonction permet d'assembler les valeurs d'un tableau en une chaîne de caractères.
MsgBox Join(Array(1, 2, 3, 4, 5), "") '=> renvoie : 12345

